پشتیبان گیر<mark>ک ، بازیابی وچک کردن پایگاه داده</mark>اSq در#C و Vb

از خواننده محترم تقاضا میشود با ارسـال نظرات و پیشـنهادات خود در مورد این مقاله ویا اشـکالات موجود در محتوی یا متن آن به آدرس <u>Mehdi.aliha@gmial.Com</u> در بهبود مقاله سـهیم شـود.

> باتشکر، مهدی محمد علیها زمستان ۱۳۸۸

بررسی اجمالی

در این سری از مقالات ، به نحوه استفاده از (Sql management Object) وانواع کارهای معمول از جمله :پشتیبان گیری،بازگرداندن، تعمیر و نگهداری ایندکسها،چک کردن وغیره که یک (DBA(Data Base Admin باید انجام دهد می پردازیم. برخی از این عملیات ها در پایگاه داده SMO نگهداری می شوند. برای دانلود این ابزار به لینک زیر مرا جعه کنید.

http://www.sqldbatips.com/showarticle.asp?ID=29

در این مقاله ما در مورد ساخت یک برنامه SMO ، وصل شدن به Sql Serve وتعیین برخی ازخواص آن وهمچنین بازیابی سرور متمرکز میشویم . نمونه کدها براساس #C و VB می باشد. وهمچنین تاکید میکنم نمونه برنامه های SMO که بخشـی از آنها مثالهای Sql Server Engineهستند را از لینک زیر دانلود کنید.

http://www.codeplex.com/MSFTEngProdSamples/Release/ProjectReleases.aspx?ReleaseId=4038

• ساختن یک برنامه SMO

برای استفاده از اشیاء SMO در یک برنامه NET. ما نیاز به ار جاء اسمبلی های SMO داریم ساده ترین راه برای این کار در ویژوال استودیو ۲۰۰۵ رفتن به مسیر project>add Reference وسپس از منوی اصلی اسمبلی های زیر را انتخاب وبه فولدر References اضافه کنید.مانند شکل زیر

- □ Microsoft.SqlServer.ConnectionInfo
- □ Microsoft.SqlServer.Smo
- □ Microsoft.SqlServer.SmoEnum
- □ Microsoft.SqlServer.SqlEnum



برای انجام وظایف خاصی امانند مدیریت سارویس ایاد اسامبلی های دیگری اضافه شاوند.اما اسامبلی های فوق شامل همه وظایف مورد نیاز ابرای انجام ابتدائی ترین مباحث می شوند. برای اطلاعات بیشتر به مرجع ابرنامه نویسای SMO در Books On Line رجوع شود.

> اگر شما دسترسی به ویژوال استودیو ندارید. میتوانید منابع مورد نیاز برنامه کاربردی خود را در کامپایلر خط فر مان اضافه کنید. بطور مثال خطوط فرمان برای کامپایلر های #C و VB بصورت زیر می باشد.

[Visual Basic]

vbc /t:exe c:\SMOtest.vb /r:

- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.ConnectionInfo.dll",
- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.Smo.dll",
- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.SmoEnum.dll",
- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.SqlEnum.dll"

[C#]

csc /t:exe /out:c:\SMOTest.exe c:\SMOtest.vb /r:

- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.ConnectionInfo.dll",
- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.Smo.dll",
- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.SmoEnum.dll",
- "C:\Program Files\Microsoft SQL Server\90\SDK\Assemblies\Microsoft.SqlServer.SqlEnum.dll"

توجه داشته باشید تمام فر مانها باید در یک خط نوشته شوند. اگرچه مثالهای متعددی برای قالب بندی خطوط وجود دارد.

• اتصال به Sql

متصل شدن به Sql Server بسیار ساده است .برای مثال اگر اتصال پیش فرض در Sql Server بصورت Windows Authentication (نحوه شناسایی) تنظیم شده باشد.ما فقط نیاز به سـاختن یک شـی جدید از شـی SMO Server داریم . همانطور که می بینید ما در مثال زیر شـروع به دریافت اطلاعاتی مانند: نام سـرور و نسـخه نگارش سـرور کردیم.

```
[Visual Basic]
```

Imports Microsoft.SqlServer.Management.Smo

Module SMOTest Sub Main()

> Dim svr As Server = New Server() Console.Writeline(svr.Name & " " & svr.Information.VersionString)

> > End Sub

End Module

[C#]

using System; using Microsoft.SqlServer.Management.Smo;

namespace SMOTest

```
{
  class Program
  {
    static void Main()
    {
      Server svr = new Server();
      Console.WriteLine(svr.Name + " " + svr.Information.VersionString);
    }
}
```

🗪 file:///C:/Documents and Settings/omid/My Documents/Visual Studio 2008/Projects/Console	- 🗆 🗙
MEHDI 8.00.194	-

برای دسترسی به سروری با نام خاص می توان نام سرور مورد نظر را به شکل زیر به سازنده Server ارسال کرد.

```
Dim svr As Server = New Server("(local)")
Server svr = new Server(@"(local)\INSTANCE01")
```

مثال

Server svr = new Server(@"(local)\SQL_SERVER");



با این حال، اگر ما بخواهیم گزینه های پیچیده تری را برای اتصال pooling که جزئی از رفتار SMO هست را کنترل کنیم. ما می توانیم از شـی ConnectionContext اسـتفاده کنیم. این شـی، فرزندی از شـی Server می باشـد که می توان خاصیت های آن را تعیین کرد.ویا با تعریف یک متغیر محلی به عنوان یک شـی ServerConnection وخاصیت های آن را تعیین وبه متد سـازنده Server ارسـال کنیم.

مثال زیر خاصیتهایی مانند: Authentication mode و UserNameوPassword را تنظیم می کند.

[Visual Basic]

Imports Microsoft.SqlServer.Management.Smo Imports Microsoft.SqlServer.Management.Common

Module SMOTest

Sub Main()

Dim svr As Server = New Server() svr.ConnectionContext.LoginSecure = false svr.ConnectionContext.Login = "username" svr.ConnectionContext.Password = "password" Console.Writeline(svr.Name & " " & svr.Information.VersionString)

End Sub

End Module

[C#]

using System; using Microsoft.SqlServer.Management.Smo; using Microsoft.SqlServer.Management.Common;

namespace SMOTest

```
{
    class Program
    {
      static void Main()
      {
        ServerConnection conn = new ServerConnection();
        conn.LoginSecure = false;
        conn.Login = "username";
        conn.Password = "password";
        Server svr = new Server(conn);
        Console.WriteLine(svr.Name + " " + svr.Information.VersionString);
      }
    }
}
```

• رفتار پیش فرض- اتصال مخلوط

در رفتار پیش فرض، SMO با استفاده از اتصال مخلوط یا Pooling، در خواست برقراری یا رها سازی اتصالات رو انجام می دهد. هرچند این رفتار می تواند توسط AutoDisconnectMode که از خاصیتهای ConnectionContext می باشد. به AutoDisconnectMode.NoAutoDisconnet یا به NonPooledConnection تغییر یابد.

اگر AutoDisconnectMode به NoAutoDisconnect تنظیم شده باشد و اتصال نیز صریح باشد، تا زمان قطع شدن اتصال Pool اتصال قابل بازگشت نیست.چون اتصال از نوع مخلوط است. اما اگر خاصیت NonPooledConnection به True تغییر یابد ماصریحاً در خواست اتصال غیر مخلوط را کرده ایم.با استفاده از Profliter در Sql Server می توان رفتار های بر نامه SMO را مشاهده نمود.

مثال

رفتار پیش فرض۔ اتصال مخلوط

Dim svr As Server = New Server() Console.Writeline(svr.Name & " " & svr.Information.VersionString)

🚰 SQL Profiler - [Untitled - 3 (BETA2)]						
讀 File Edit View Replay Tools Window Help _ 문자						
EventClass	SP	TextData	NTUser	ClientProcess	App	
Audit Login	53	network protocol: LPC set quoted	Administ	3324	SQL	
RPC:Completed	53	exec sp_resetconnection	Administ	3324	SQL	
SQL:BatchCompleted	53	select SERVERPROPERTY(N'serve	Administ	3324	SQL	
RPC:Completed	53	exec sp_resetconnection	Administ	3324	SQL	
SQL:BatchCompleted	53	SELECT SERVERPROPERTY(N'Pr	Administ	3324	SQL	
Audit Logout	53		Administ	3324	SQL 🚽	
Trace is stopped. Ln 15, Col 1 Rows: 15						

Dim svr As Server = New Server() svr.ConnectionContext.NonPooledConnection = True svr.ConnectionContext.Connect() Console.Writeline(svr.Name & " " & svr.Information.VersionString) svr.ConnectionContext.Disconnect()

	SQL Profiler - [Untitled - 3 (BETA2)]					
<u>k</u>	🚡 File Edit View Replay Tools Window Help					
	EventClass	SP	TextData	NTUser	ClientP 🔺	
	Audit Login	53	network protocol: LPC set quoted_identifier on	Administ	3252	
	SQL:BatchCompleted	53	select SERVERPROPERTY(N'servername')	Administ	3252	
	SQL:BatchCompleted	53	SELECT SERVERPROPERTY(N'ProductVersion')	Administ	3252 📃	
	Audit Logout	53		Administ	3252 💌	
Trace is stopped. Ln 13, Col 1 Rows: 13						

• پشتیبان گیری DataBase

برای تهیه یک نسخه پشتیبان مانیاز به دوشی داریم .۱- شی Server ۲- شی Backup

در ساده ترین شـکل مانیاز به یک شـی Backup داریم .که قبل از فراخوانی متد SqlBackup باید خواص شـی Backup را تنظیم و شـی Server را به آن ارسـال کنیم. در مثال زیر یک پشتیبانی کامل از پایگاه داده SMO درمسیر SMOTest.bak گرفته شـده اسـت.

[Visual Basic]

Imports Microsoft.SqlServer.Management.Smo

Module SMOTest

Sub Main() Dim svr As Server = New Server() Dim bkp As Backup = New Backup() bkp.Devices.AddDevice("C:\SMOTest.bak", DeviceType.File) bkp.Database = "SMO" bkp.Action = BackupActionType.Database bkp.Initialize = True bkp.PercentCompleteNotification = 10 AddHandler bkp.PercentComplete, AddressOf ProgressEventHandler bkp.SqlBackup(svr)

End Sub

Private Sub ProgressEventHandler(ByVal sender As Object, _ ByVal e As PercentCompleteEventArgs) Console.WriteLine(e.Percent.ToString + "% backed up") End Sub

End Module

[C#]

using Microsoft.SqlServer.Management.Smo;

```
namespace SMOTest
{
  class Program
  {
     static void Main()
     {
        Server svr = new Server();
        Backup bkp = new Backup();
        bkp.Devices.AddDevice(@"C:\SMOTest.bak", DeviceType.File);
        bkp.Database = "SMO";
        bkp.Action = BackupActionType.Database;
        bkp.Initialize = true;
        bkp.PercentCompleteNotification = 10;
        bkp.PercentComplete += new
        PercentCompleteEventHandler(bkp_PercentComplete);
        bkp.SqlBackup(svr);
     }
     static void bkp_PercentComplete(object sender, PercentCompleteEventArgs e)
        Console.WriteLine(e.Percent.ToString() + "% backed up");
     }
  }
}
```

در این تمرین از یک Event Handler در متد _bkp_PercentComplete استفاده شده که وضعیت پشتیبان گیری را نمایش می دهد.

خصوصیات زیادی برای تنظیم شـی Backup وجود دارد. برای دیدن لیسـت تمام این خصوصیات می توانید به books on line مراجعه کنید.(اگر شما می خواهید کتاب های رایگان Sql 2005 را نصب کنید به لینک زیر مراجعه کنید.)

http://www.microsoft.com/downloads/details.aspx?familyid=BE6A2C5D-00DF-4220-B133-29C1E0B6585F&displaylang=en

• پشتیبان گیری بصورت Filegroup

در مثال زیر از WorkOrderGroup که در پایگاه داده Adventureworks قرار دارد بصورت filegroup پشتیبان گیری شده است.(اسکریپت تبدیل پایگاه داده مجازی Adventureworks به چندین filegroup را می توان در مثالهای SQL Server Engine پیدا کرد مسیر پیش فرض قرار گیری مثالها (C:\Program Files\Microsoft SQL Server\90\Samples\Engine\Administration\FileGroups\Scripts)می باشد.

Backup ها دارای Backup های خاصی هستند که بصورت اشیاء Backup به مجموعه DatabaseFileGroups اضافه میشوند. به منظور تهیه Backup های خاص .می توان آنها را به مجموعه DatabaseFileGroup اضافه نمود.

[Visual Basic]

Imports Microsoft.SqlServer.Management.Smo

```
Module SMOTest

Sub Main()

Dim svr As Server = New Server()

Dim bkp As Backup = New Backup()

bkp.Database = "AdventureWorks"

bkp.Action = BackupActionType.Files

bkp.DatabaseFileGroups.Add("WorkOrderGroup")

bkp.Devices.AddDevice("C:\AWFGTest.bak", DeviceType.File)

bkp.SqlBackup(svr)

End Sub

End Module
```

[C#]

using Microsoft.SqlServer.Management.Smo;

```
namespace SMOTest
{
  class Program
  {
   static void Main()
   {
   Server svr = new Server();
   Backup bkp = new Backup();
   bkp.Database = "AdventureWorks";
   bkp.Action = BackupActionType.Files;
   bkp.DatabaseFileGroups.Add("WorkOrderGroup");
   bkp.Devices.AddDevice(@"C:\AWFGTest.bak", DeviceType.File);
   bkp.SqlBackup(svr);
   }
 }
}
```

پشـتیبان گیری بصورت Differential یا دیفرانسـیل

تهیه یک نسخه پشتیبان بصورت Differential بسیار ساده است . فقط کافیست خاصیت Incremental از شـی Backup را به True تنظیم کنیم.

bkp.Incremental = True

پشتیبان گیری بصورت Log

پشتیبان گیری Log بسیار ساده است .فقط کافیست خاصیت Action از شری Backup را به BackupActionType.Log تنظیم کنیم.

[Visual Basic]

```
Imports Microsoft.SqlServer.Management.Smo
```

```
Module SMOTest

Sub Main()

Dim svr As Server = New Server()

Dim bkp As Backup = New Backup()

bkp.Action = BackupActionType.Log

bkp.Database = "SMO"

bkp.Devices.AddDevice("c:\SMOTest.trn", DeviceType.File)

bkp.SqlBackup(svr)

End Sub
```

End Module

[C#]

using Microsoft.SqlServer.Management.Smo;

```
namespace SMOTest
{
    class Program
    {
      static void Main()
      {
        Server svr = new Server();
        Backup bkp = new Backup();
        bkp.Action = BackupActionType.Log;
        bkp.Database = "SMO";
        bkp.Devices.AddDevice(@"c:\SMOTest.trn", DeviceType.File);
        bkp.SqlBackup(svr);
    }
}
```

• بازیابی DataBase

برای بازیابی یک DataBase مانیاز به دوشتی داریم .۱۰ شنی Server ۲- شنی Restore

در ساده ترین شـکل مانیاز به یک شـی Restore داریم .که قبل از فراخوانی متد SqlRestore باید خواص شـی Restore را تنظیم و شـی Server را به آن ارسـال کنیم. در مثال زیر یک بازیابی کامل از پایگاه داده SMO درمسیر SMOTest.bak جایگزین پایگاه داده فعلی می شود.

[Visual Basic]

Imports Microsoft.SqlServer.Management.Smo

```
Module SMOTest
  Sub Main()
     Dim svr As Server = New Server()
     Dim res As Restore = New Restore()
     res.Devices.AddDevice("C:\SMOTest.bak", DeviceType.File)
     res.Database = "SMO"
     res.ReplaceDatabase = True
     res.PercentCompleteNotification = 10
     AddHandler res.PercentComplete, AddressOf ProgressEventHandler
     res.SqlRestore(svr)
  End Sub
  Private Sub ProgressEventHandler(ByVal sender As Object, _
                        ByVal e As PercentCompleteEventArgs)
     Console.Writeline(e.Percent.ToString + "% restored")
  End Sub
End Module
```

[C#]

using System; using Microsoft.SqlServer.Management.Smo;

```
namespace SMOTest
{
           class Program
           {
                      static void Main()
                      {
                                 Server svr = new Server();
                                 Restore res = new Restore();
                                 res.Database = "SMO";
                                 res.Action = RestoreActionType.Database;
                                 res.Devices.AddDevice(@"C:\SMOTest.bak", DeviceType.File);
                                 res.PercentCompleteNotification = 10;
                                 res.ReplaceDatabase = true;
                                 res.PercentComplete += new
               PercentCompleteEventHandler(ProgressEventHandler);
                                 res.SqlRestore(svr);
                      }
     static void ProgressEventHandler(object sender, PercentCompleteEventArgs e)
                      {
                                 Console.WriteLine(e.Percent.ToString() + "% restored");
                      }
  }
}
              در این تمرین از یک Event Handler در متد ProgressEventHandler استفاده شده که وضعیت پشتیبان گیری را نمایش می دهد.
```

خصوصیات زیادی برای تنظیم شـی Restore وجود دارد. برای دیدن لیسـت تمام این خصوصیات می توانید به books on line مراجعه کنید.(اگر شـما می خواهید کتاب های رایگان Sql 2005 را نصب کنید به لینک زیر مراجعه کنید.)

http://www.microsoft.com/downloads/details.aspx?familyid=BE6A2C5D-00DF-4220-B133-29C1E0B6585F&displaylang=en

• بازیابی Database در یک محل جدید

با استفاده از معادل دستورات T-SQL WITH MOVE برای بازیابی وهمچنین تنظیم خاصیتهای RelocateFiles از شبی Restore

وتعیین پارامتر های شـی RelocateFile به کمک SMO این عمل را انجام می دهیم. در مثال زیر ما یک کپی از پایگاه داده ها و فایلهای Log در پایگاه داده SMO2 بازیابی کردیم.

> در سـازنده RelocateFile از دو پارامتر اسـتفاده شـده :۱- نام منطقـی فایل ۲- نام فیزیکی فایل . که نقش آن فراهم کردن محلی برای انتقال فایل ها در زمان بازیابی اسـت.

[Visual Basic]

Imports Microsoft.SqlServer.Management.Smo

```
Module SMOTest

Sub Main()

Dim svr As Server = New Server()

Dim res As Restore = New Restore()

res.Devices.AddDevice("C:\SMOTest.bak", DeviceType.File)

res.Database = "SMO2"

res.NoRecovery = False

res.ReplaceDatabase = True

res.RelocateFiles.Add(New RelocateFile("SMO", "c:\SMO2.mdf"))

res.RelocateFiles.Add(New RelocateFile("SMO_Log", "c:\SMO2.ldf"))

res.SqlRestore(svr)

End Sub

End Module
```

```
[C#]
```

using Microsoft.SqlServer.Management.Smo;

```
namespace SMOTest
{
  class Program
  {
     static void Main()
     {
        Server svr = new Server();
        Restore res = new Restore();
        res.Database = "SMO2";
        res.Action = RestoreActionType.Database;
        res.Devices.AddDevice(@"C:\SMOTest.bak", DeviceType.File);
       res.ReplaceDatabase = true;
        res.RelocateFiles.Add(new RelocateFile("SMO", @"c:\SMO2.mdf"));
        res.RelocateFiles.Add(new RelocateFile("SMO_Log", @"c:\SMO2.Idf"));
       res.SqlRestore(svr);
     }
  }
}
```

اطلاعاتی در مورد فایلهای پشتیبان

تعدادی متد در شـی Restore وجود دارد که با استفاده از هر کدام می تواند اطلاعاتی در مورد درایو Backup و فایلهای Backup نشـان داد. که این متد ها عبارتند از : ۱- ReadMediaHeader ۲۰ ReadFileList ۲۰ ReadBackupHeader .

در مثال زیر ما با استفاده از متد ReadFileLis لیستی از نام فایلهای منطقی در درایو c:\SMOTest.bak را در کنسول نمایش می دهیم.

[Visual Basic]

```
Imports Microsoft.SqlServer.Management.Smo
Imports System.Data
```

```
Module SMOTest
Sub Main()
```

Dim svr As Server = New Server() Dim res As Restore = New Restore() Dim dt As DataTable Dim foundrows As DataRow()

res.Devices.AddDevice("C:\SMOTest.bak", DeviceType.File)
dt = res.ReadFileList(svr)

foundrows = dt.Select(Nothing)

```
For Each dr As DataRow In foundrows
Console.WriteLine(dr("LogicalName").ToString())
Next
End Sub
End Module
```

[C#]

```
using System;
using System.Data;
using Microsoft.SqlServer.Management.Smo;
```

```
namespace SMOTest
```

```
{
  class Program
  {
    static void Main()
    {
      Server svr = new Server();
      Restore res = new Restore();
      DataTable dt;
      DataRow[] foundrows;
      res.Devices.AddDevice(@"C:\SMOTest.bak", DeviceType.File);
      dt = res.ReadFileList(svr);
      foundrows = dt.Select();
      foreach (DataRow r in foundrows)
      Console.WriteLine(r["LogicalName"].ToString());
      }
    }
 }
}
```

از آنجا که همه این متد ها شـی از نوع DataTable بر می گردانند . می توان به راحتی به همه سـتون های اسـتخراج شـده دسـترسـی داشـت. مثال زیر، مجموعه ای از سـتونهای اسـتخراج شـده از شـی DataTable را نمایش می دهد. [Visual Basic] Imports Microsoft.SqlServer.Management.Smo Imports System.Data

```
Module SMOTest
  Sub Main()
     Dim svr As Server = New Server()
     Dim res As Restore = New Restore()
     Dim dt As DataTable
     Dim foundrows As DataRow()
     res.Devices.AddDevice("C:\SMOTest.bak", DeviceType.File)
     dt = res.ReadFileList(svr)
     foundrows = dt.Select(Nothing)
     For Each col As DataColumn In dt.Columns
        Console.Write(col.ColumnName + vbTab)
     Next
     Console.WriteLine()
     For Each dr As DataRow In foundrows
        For Each col As DataColumn In dt.Columns
          Console.Write(dr(col.ColumnName).ToString() + vbTab)
        Next
        Console.WriteLine()
     Next
  End Sub
End Module
```

```
• بازیابی Log
```

بازیابی Log بسیار ساده است . فقط کافی است خاصیت Action از شـی Restore را به RestoreActionType.Log تنظیم کنیم .خاصیت های دیگر برای پشتیبان گیری Log وجود دارند مثل ToPointInTime که با استفاده از آن می توان عمل بازیابی را به نقطه خاصی از زمان معطوف کرد.

[Visual Basic] Imports Microsoft.SqlServer.Management.Smo

```
Module SMOTest

Sub Main()

Dim svr As Server = New Server()

Dim res As Restore = New Restore()

res.Devices.AddDevice("C:\SMOTest.trn", DeviceType.File)

res.Database = "SMO"

res.NoRecovery = False

res.Action = RestoreActionType.Log

res.SqlRestore(svr)

End Sub

End Module
```

[C#]

using System; using Microsoft.SqlServer.Management.Smo;

```
namespace SMOTest
{
  class Program
  {
    static void Main()
    {
    Server svr = new Server();
    Restore res = new Restore();
    res.Database = "SMO";
    res.Action = RestoreActionType.Log;
    res.Devices.AddDevice(@"C:\SMOTest.trn", DeviceType.File);
            res.NoRecovery = false;
    res.SqlRestore(svr);
    }
 }
}
```

• چک کردن درستی DataBase

برای چک کردن درستی DataBase از متد های SMOاستفاده می شود.

در جدول زیر متد های SMO همراه با دستورات معادل TSQL آن را می بینید.

SMO Method	TSQL Command
CheckAllocations(RepairType.None)	DBCC CHECKALLOC WITH NO_INFOMSGS
CheckAllocationsDataOnly()	DBCC CHECKALLOC(N'databasename', NOINDEX)
CheckCatalog()	DBCC CHECKCATALOG
CheckTables(RepairType.None)	DBCC CHECKDB WITH NO_INFOMSGS
CheckTablesDataOnly()	DBCC CHECKDB(N'databasename', NOINDEX)

همه متدها در جدول بالا یک مقدار StringCollection برمی گردانند. ما میتوانیم با تکرار این عمل نتایج برگشتی راچک کنیم تا از درستی بانک مطمئن شویم.

[Visual Basic]

Imports Microsoft.SqlServer.Management.Smo Imports System.Collections.Specialized

```
Module SMOTest
```

```
Sub Main()
Dim svr As Server = New Server()
Dim sc As StringCollection
Dim db As Database
db = svr.Databases("AdventureWorks")
```

```
sc = db.CheckTables(RepairType.None)
```

```
For c As Integer = 0 To (sc.Count - 1)
If sc(c).Length > 0 Then
Console.WriteLine(sc(c))
End If
Next
```

End Sub

End Module

```
[C#]
```

```
using System;
using Microsoft.SqlServer.Management.Smo;
using System.Collections.Specialized;
```

```
namespace SMOTest
```

```
{
  class Program
   {
     static void Main()
      {
        Server svr = new Server();
        Database db;
        db = svr.Databases["AdventureWorks"];
        StringCollection sc;
        sc = db.CheckCatalog();
        for (int i = 0; i < \text{sc.Count}; i + +)
        {
                  if (sc[i].Length > 0)
          Console.WriteLine(sc[i]);
        }
     }
  }
}
```

• شناسایی Backup ها

برای شناسایی یک Backup مانیاز به دوشی داریم .۱۰ ش.ی Server ۲- شی Restore

در ساده ترین شکل مانیاز به یک شی Restore داریم .که قبل از فراخوانی متد SqlVerify باید خواص شی Restore را تنظیم و شی Server را به آن ارسال کنیم. در مثال زیر یک شناسایی ازیک Backup پایگاه داده در مسیر SMOTest.bak (انجام شده. در این مثال متد SqlVerify سربارگذاری گردیده.بطوری که در صورت بر گشت مقدار False از متد SqlVerify خطای Verify آن نمایش داده میشود.

```
[Visual Basic]
```

Imports Microsoft.SqlServer.Management.Smo

```
Module SMOTest
Sub Main()
Dim svr As Server = New Server()
Dim res As Restore = New Restore()
Dim verified As Boolean
Dim errormsg As String = ""
```

res.Devices.AddDevice("C:\SMOTest.bak", DeviceType.File)
verified = res.SqlVerify(svr, errormsg)

```
Console.WriteLine("verified = " & verified.ToString())
```

```
If verified = False Then
Console.WriteLine(errormsg)
End If
```

End Sub End Module

```
[C#]
```

```
using System;
using Microsoft.SqlServer.Management.Smo;
```

```
namespace SMOTest
```

```
Restore res = new Restore();
Boolean verified;
String errormsg;
```

res.Devices.AddDevice(@"C:\SMOTest.bak", DeviceType.File);

verified = res.SqlVerify(svr, out errormsg);

Console.WriteLine("verified = " + verified.ToString());

```
if (verified == false)
    Console.WriteLine(errormsg);
```

```
}
}
```

Mehdi.aliha@gmial.Com